

එදිනෙදා ජීවිතයට

# ක්ෂුද්‍ර පාලන ඒකක




## MICROCONTROLLERS

හයවන ලිපිය - හයවන කොටස (VI)

### බාහිර අතුරුබිඳුම් භාවිත කරන අයුරු

PIC 16F84(A) මයික්‍රොක්‍රොලරයට පිටතින් අතුරුබිඳුමක් ලබා දීම හා ඊට අනුරූප ව ක්‍රමලේඛනය සකස් කරගත යුතු ආකාරය මෙම ලිපියෙන් විස්තර කෙරේ. 6.2 ලිපියේ පළ වූ පරිපථයට බොහෝ දුරට සමාන පරිපථයක් මෙහි දී ද යොදාගෙන ඇති අතර රූපසටහන අංක 1න් අදාළ පරිපථ සටහන දක්වා ඇත. එ අනුව SW1 බොත්තම වරක් එබූ විට සථන බණ්ඩ ප්‍රදර්ශකයේ දිස් වන අගය 1කින් වැඩි විය යුතු ය. ඊට අදාළ ක්‍රමලේඛනය රූපසටහන අංක 2න් දැක්වේ.

ක්‍රමලේඛනය ආරම්භයේ දී ම අදාළ රෙජිස්ටරය හා එවැනි පිහිටුම් අංක දක්වා ඇති අතර ඉන් පසුව ඇති orgo මගින් MPLAB මෘදුකාංගයට gotto main උපදෙස ලිවිය යුතු ස්ථානය දක්වයි. Reset vector යනුවෙන් හැඳින්වෙන්නේ මයික්‍රොකොන්ට්‍රොලරයේ ක්‍රමලේඛනය ගබඩා කරන කොටසේ (Program memory) ඇති මුළු ම පිහිටුමයි. එහි පිහිටුම් අංකය 0 වන අතර විදුලිය සැපයූ විගස මෙම ස්ථානයේ ලියා ඇති goto main උපදෙස ක්‍රියාත්මක කරයි. එ අනුව ක්‍රමලේඛනයේ main නැමැති ස්ථානයට ගමන් කර ඉන්පසුව ඇති උපදෙස් එකින් එක ක්‍රියාත්මක කරයි. එම උපදෙස් අතුරින්



රූපසටහන අංක 1

movwf INTCONහි INTCON රෙජිස්ටරය පිළිබඳව යමක් කිව යුතු ය. INTCON රෙජිස්ටරයට ලියන අගයන් අනුව සමහර අතුරුබිඳුම් ක්‍රියාත්මක වීම හෝ නො වීම තීරණය වේ. INTCON යනු Interrupt Configuration Registerහි කෙටි යෙදුම වේ. මෙම රෙජිස්ටරයේ එක් එක් බිටුවලින් කෙරෙන කාර්යයන් රූප සටහන අංක 3න් දැක්වේ. Display number 0 යන සටහනට පසුව ඇති උපදෙස් දෙකින් කෙරෙනුයේ ආරම්භයේ දී ම සථන බණ්ඩ ප්‍රදර්ශකයේ බිංදුව ප්‍රදර්ශනය වීමට සැලැස්වීමයි. ඉන්පසුව ඇති Loop1 goto Loop1 උපදෙසට අනුව එම උපදෙසට ම නොනවත්වා දිගට ම ක්‍රියාත්මක කරයි.

මේ අතර SW1 බොත්තම එබූව හොත් එ මගින් බාහිර අතුරුබිඳුමක් ජනනය කරයි. එවිට Loop1 goto Loop1 උපදෙස දිගට ම ක්‍රියාත්මක කිරීම නවතා ක්‍රමලේඛනයේ ගබඩා කර ඇති මතකයේ 4 වැනි පිහිටුමට ගමන් කරයි. org 40 පසුව ලියා ඇත්තේ එවිට ක්‍රියාත්මක කළ යුතු උපදෙස් වේ. මෙම උපදෙස් හොනුවේ මුලින් ම ඇති 6tfss INTCON,1 යනු bit test f Skip if set යන්නයි. මෙහි දී f යනු INTCON රෙජිස්ටරය වන අතර සත්සන්දනය කළ යුතු බිටුව වනුයේ 1 වන බිටුවයි. RBO අග්‍රයෙන් බාහිර අතුරුබිඳුමක් ජනනය වූ විගස මෙම බිටුව 1 බවට පත් වේ. එය වී ඇත්නම්,

වැනි පිටුවට මාරුව වියව විද්‍යාලයේ විද්‍යුත් හා විදුලි සංදේශ අංශයේ ගාමිණී ජයසිංහ/කෝලින ධර්මප්‍රිය

REGISTER 2-3: INTCON REGISTER (ADDRESS 0Bh, 8Bh)

R/W-0	R/W-0	R/W-0	R/W-0	R/W-0	R/W-0	R/W-0	R/W-x
GIE	EEIE	TOIE	INTE	RBIE	TOIF	INTF	RBIF
bit 7				bit 0			

- bit 7

GIE: Global Interrupt Enable bit

1 = Enables all unmasked interrupts

0 = Disables all interrupts
- bit 6

EEIE: EE Write Complete Interrupt Enable bit

1 = Enables the EE Write Complete interrupts

0 = Disables the EE Write Complete interrupt
- bit 5

TOIE: TMR0 Overflow Interrupt Enable bit

1 = Enables the TMR0 interrupt

0 = Disables the TMR0 interrupt
- bit 4

INTE: RB0/INT External Interrupt Enable bit

1 = Enables the RB0/INT external interrupt

0 = Disables the RB0/INT external interrupt
- bit 3

RBIE: RB Port Change Interrupt Enable bit

1 = Enables the RB port change interrupt

0 = Disables the RB port change interrupt
- bit 2

TOIF: TMR0 Overflow Interrupt Flag bit

1 = TMR0 register has overflowed (must be cleared in software)

0 = TMR0 register did not overflow
- bit 1

INTF: RB0/INT External Interrupt Flag bit

1 = The RB0/INT external interrupt occurred (must be cleared in software)

0 = The RB0/INT external interrupt did not occur
- bit 0

RBIF: RB Port Change Interrupt Flag bit

1 = At least one of the RB7:RB4 pins changed state (must be cleared in software)

0 = None of the RB7:RB4 pins have changed state

Legend:			
R = Readable bit	W = Writable bit	U = Unimplemented bit, read as '0'	
- n = Value at POR	'1' = Bit is set	'0' = Bit is cleared	x = Bit is unknown

රූපසටහන අංක 3

```
;****Define Registers****
PC equ 02h
STATUS equ 03h
PORTB equ 06h
TRISB equ 86h
INTCON equ 0Bh
CounterL equ 0Dh
CounterH equ 0Eh
Count equ 0Fh

;****Reset vector****
Org 0
goto main ;1st instruction ;at 0th location

;****Interrupt vector****
org 4
btfss INTCON,1;RB0 interrupt?
goto INTend ;If no return

incf Count,1 ;If yes count up
movlw D'10' ;load w with ten
xorwf Count,0 ;compare Count & 10

btfsc STATUS,2;if equal result 0
clrf Count ;Then clear Count
movf Count,0 ;Else continue
call Table ;Look Up Table
movwf PORTB ;Writ the number

Loop decfsz CounterL,1
goto Loop
decfsz CounterH,1
goto Loop

bcf INTCON,1;Clear the interrupt flag

INTend retfie ;Return from interrupt

;****Main Program start from here****
org D'25' ;start writing from ;25th location

;****Set up the Ports****
main bsf STATUS,5;Switch to Bank 1
clrf TRISB ;PORT B output
bsf TRISB,0 ;RB0 input
bcf STATUS,5;Switch to Bank 0
movlw b'10010000';Enable RB0 and
movwf INTCON ;external interrupt
clrf Count ;Interrupt count=0

;****Display number 0****
movlw b'01111111'
movwf PORTB ;Writ the number 0

;****Loop forever****
Loop1 goto Loop1 ;If no then wait

;****Look Up Table for bit patterns****
Table addwf PC,1
retlw b'01111111';Number 0
retlw b'00001101';Number 1
retlw b'10110111';Number 2
retlw b'10011111';Number 3
retlw b'11001101';Number 4
retlw b'11011011';Number 5
retlw b'11111011';Number 6
retlw b'00001111';Number 7
retlw b'11111111';Number 8
retlw b'11001111';Number 9

end
```

රූපසටහන අංක 2